Make Model Year
The Make-Model-Year code customization shows how to display a set of dropdown lists that update themselves based on the values selected in other dropdown lists.  This example has become to be called the Make-Model-Year example based on the ability to select the Make of the car (e.g., Toyota), followed by a specific model for that make (e.g., Camry), followed by the specific year that make and model were offered (e.g., 2008).

When the end user selects a value in the first dropdown (Make), the second dropdown is populated based on the first value selected. This process is repeated for subsequent dropdowns.
Database and Tables

The following three tables make up this example. 
Owner table in TrainingSouthwind.mdb

· Id: An AutoNumber field that is a unique Id.

· OwnerName: The name of the owner.

· Address: Address of the owner.
Vehicle Type table:

· Id: An AutoNumber field that is a unique Id.

· Make: The vehicle Make (e.g., Toyota).

· Model: The vehicle Model (e.g., Corolla).

· Year: The vehicle Year (e.g., 2008)

Owner Vehicle table:

· OwnerId: The Owner Id

· VehicleTypeId: The Id of the Vehicle Type.

The Owner table is considered the master table, and the Owner Vehicle table will be the detail table and list all of the vehicles owned by the owner.    

When a New Owner is being added or an existing Owner is being edited, the end user will have a chance to add, edit or delete a vehicle.  Normally, Iron Speed Designer creates a drop-down list based on the relationship between the VehicleTypeId in the OwnerVehicle table with the Id field of the VehicleType table.  This relationship results in a Display Foreign Key As (DFKA) field which by default is the first string field in the VehicleType table.  This field would normally be the Make field.  
This code customization will demonstrate how to change the Add or Edit page to display the Make, Model and Year dropdowns instead of just one dropdown.  When the Make is selected, the Model will become enabled.  When the Model is selected, the Year will become enabled.  If the Make is changed, the Model and Year must be blanked out and disabled until the Make is selected again.
Application

The Training1 application contains the following four pages for the Owner table:
· Add record

· Edit record

· Show record

· Show table

The default master-detail pages created by Iron Speed Designer will show a single dropdown for the Vehicle Type in the detail section of the page.  This dropdown is automatically mapped to the Make field of the Vehicle Type table.  Since there are multiple vehicles with the same make, the initial dropdown created by Iron Speed Designer will list duplicate values.

This code customization will change this dropdown to list unique values, and add two more dropdowns for the model and year.  The saved value in database will only correspond to the specific combination of make, model and year.

Customizations

Add Make and Model Dropdown Lists
By default, a GEN:FieldValue code generation tag is created for VehicleTypeId in the detail portion of the Add Owner page.  Add two additional GEN:FieldValue tags before the VehicleTypeId tag for Make and Model as shown below:

<td class="ttc" style="white-space:nowrap;">
  <GEN:FieldValue NAME="Make" />&nbsp;
  <GEN:FieldValue NAME="Model" />&nbsp;
  <GEN:FieldValue NAME="VehicleTypeId" />
</td>

Right-mouse click the VehicleTypeId and rename it to Year

Bring up the Page Properties dialog box and bind the Make and Model tags to the Vehicle Type Id field.

Repeat this for the Model tag added earlier as well.

Change the Display Type of both Make and Model to a dropdown list:

Build the application to create all of the code for each of the controls.
Code Customizations
There are three dropdowns for each row:

1. Make

2. Model

3. Year
These three dropdowns are implemented in the Table Control Row class called BaseOwnerVehicleTableControlRow.  This class corresponds to each row in the table.  The OwnerVehicleTableControlRow class derives from the BaseOwnerVehicleTableControlRow and can override any of the functions in lower level classes.
Setting AutoPostBack
The first step is to set the AutoPostBack attribute for both the Make and the Model dropdowns so that if the end user changes the value in the dropdown, the Selected Index Changed method will be called so we can populate the other dropdowns.  We do not need to set the AutoPostBack for the last dropdown since changes to the values of that dropdown do not have an impact on any other dropdown.
This is accomplished by handling the MyBase.Init event and setting the AutoPostBack to True for both Make and Model.   This is followed by adding a handler to handle the Selected Index Changed event for these two dropdowns.  Finally, we want to enable the Make dropdown and disable the other two dropdowns.

Private Sub MultipleDropdown_myInit(ByVal sender As Object, ByVal e As System.EventArgs) _


Handles MyBase.Init

    Make.AutoPostBack = True

    Model.AutoPostBack = True

    AddHandler Make.SelectedIndexChanged, AddressOf Make_SelectedIndexChanged

    AddHandler Model.SelectedIndexChanged, AddressOf Model_SelectedIndexChanged

    Make.Enabled = True

    Model.Enabled = False

    Year.Enabled = False

End Sub
Selected Index Changed for Make Dropdown
The Selected Index Changed for the Make dropdown will simply enable the Model dropdown and call its Populate method.

Protected Overrides Sub Make_SelectedIndexChanged(ByVal sender As Object, ByVal e As EventArgs)

    Model.Enabled = True

    ' Model DropDownList will display 100 items.

    ' You can set the number of items displayed in the DropDownList.

    Me.PopulateModelDropDownList(Nothing, 100)

End Sub

Selected Index Changed for Model Dropdown

The Selected Index Changed for the Model dropdown will simply enable the Year (Year) dropdown and call its Populate method.

Protected Overrides Sub Model_SelectedIndexChanged(ByVal sender As Object, ByVal e As EventArgs)

    Year.Enabled = True

    ' Model DropDownList will display 100 items.

    ' You can set the number of items displayed in the DropDownList.

    Me.PopulateYearDropDownList(Nothing, 100)

End Sub

Fill the Make Dropdown

The Make dropdown needs to be filled with unique Make names from the VehicleType table.  This function will be called with the currently selected value if one is selected, and the maximum number of items to display in the dropdown.  By default, the maximum number of items are 100, but can be changed on the Application Wizard Options or on the Attributes tab of the Page Properties dialog box for the Make dropdown control.

We will repeatedly call the GetRecords function on the VehicleTypeTable to retrieve maxItems (100) records at a time.  We will create a ListItem for each record from the database and add it to the dropdown list if it is not already in the list.  Note that the displayed text and the selected value will both be set to the actual make of the vehicle.  This value will be used later in the Populate Model dropdown to lookup the models for the selected make.

Finally, the selected value is selected in the list followed by adding the generic Please Select item at the top.

Protected Overrides Sub PopulateMakeDropDownList( _


ByVal selectedValue As String, _


ByVal maxItems As Integer)

    'Step 1: Setup the WHERE clause.
    Dim wc As WhereClause = CreateWhereClause_MakeDropDownList()

    Dim orderBy As OrderBy = New OrderBy(False, True)

    orderBy.Add(VehicleTypeTable.Make, OrderByItem.OrderDir.Asc)

    Me.Make.Items.Clear()

    Dim rec As VehicleTypeRecord

    Dim pageNum As Integer = 0
    Dim recs As VehicleTypeRecord()

    ' Step 2: Read from the database and fill the Make dropdown

' until either there are no more records in the DB

' or until we have reached the maxItems
    Do
        recs = VehicleTypeTable.GetRecords(wc, orderBy, pageNum, maxItems)

        For Each rec In recs

            ' Create the item and add to the list.
            Dim item As ListItem = New ListItem(rec.Make, rec.Make)

            ' If the item is already in the list, do not add it.
            If Not Me.Make.Items.Contains(item) Then
                Me.Make.Items.Add(item)

            End If
        Next
        pageNum += 1
    Loop While recs.Length = maxItems AndAlso Me.Make.Items.Count < maxItems

    ' Step 3: Setup the selected item.
    If Not selectedValue Is Nothing AndAlso _




selectedValue.Trim <> "" AndAlso _




Not SetSelectedValue(Me.Make, selectedValue) AndAlso _




Not MiscUtils.SetSelectedValue(Me.Make, OwnerVehicleTable.VehicleTypeId.Format(selectedValue)) Then
        Dim fvalue As String = OwnerVehicleTable.VehicleTypeId.Format(selectedValue)

        Dim item As ListItem = New ListItem(fvalue, selectedValue)

        item.Selected = True
        Me.Make.Items.Insert(0, item)

    End If
    Me.Make.Items.Insert(0, New ListItem(Page.GetResourceValue("Txt:PleaseSelect", "Training1"), _

          "--PLEASE_SELECT--"))

    ' Step 4: Disable & clear other dropdowns
    Me.Model.Items.Clear()

    Me.Model.Enabled = False
    Me.Year.Items.Clear()

    Me.Year.Enabled = False
End Sub
Fill the Model Dropdown

The Model dropdown needs to be filled with unique Model names from the VehicleType table.  This function is similar to the Populate Make except with one difference.  The currently selected Make is added to the Where Clause before retrieving new records from the database.

Note that the ListItem created for adding to the dropdown will contain the model name for the displayed text and the selected value.  This value will be used later in the Populate Vehicle Id dropdown to lookup the Year for the selected make and model.

Protected Overrides Sub PopulateModelDropDownList( _


ByVal selectedValue As String, _


ByVal maxItems As Integer)

    'Step 1: Setup the WHERE clause.
    Dim wc As WhereClause = CreateWhereClause_ModelDropDownList()

    wc.iAND(VehicleTypeTable.Make, BaseFilter.ComparisonOperator.EqualsTo, Me.Make.SelectedValue)

    Dim orderBy As OrderBy = New OrderBy(False, True)

    orderBy.Add(VehicleTypeTable.Model, OrderByItem.OrderDir.Asc)

    Me.Model.Items.Clear()

    Dim rec As VehicleTypeRecord

    Dim pageNum As Integer = 0
    Dim recs As VehicleTypeRecord()

    ' Step 2: Read from the database and fill the Make dropdown

' until either there are no more records in the DB

' or until we have reached the maxItems

Do
        recs = VehicleTypeTable.GetRecords(wc, orderBy, pageNum, maxItems)

        For Each rec In recs

            ' Create the item and add to the list.
            Dim item As ListItem = New ListItem(rec.Model, rec.Model)

            ' If the item is already in the list, do not add it.
            If Not Me.Model.Items.Contains(item) Then
                Me.Model.Items.Add(item)

            End If
        Next
        pageNum += 1
    Loop While recs.Length = maxItems AndAlso Me.Model.Items.Count < maxItems

    ' Step 3: Setup the selected item.
    If Not selectedValue Is Nothing AndAlso _




selectedValue.Trim <> "" AndAlso _




Not SetSelectedValue(Me.Model, selectedValue) AndAlso _




Not MiscUtils.SetSelectedValue(Me.Model, OwnerVehicleTable.VehicleTypeId.Format(selectedValue)) Then
        Dim fvalue As String = OwnerVehicleTable.VehicleTypeId.Format(selectedValue)

        Dim item As ListItem = New ListItem(fvalue, selectedValue)

        item.Selected = True
        Me.Model.Items.Insert(0, item)

    End If
    Me.Model.Items.Insert(0, New ListItem(Page.GetResourceValue("Txt:PleaseSelect", "Training1"), _

          "--PLEASE_SELECT--"))

    ' Step 4: Disable & clear other dropdowns
    Me.Year.Items.Clear()

    Me.Year.Items.Insert(0, New ListItem(Page.GetResourceValue("Txt:PleaseSelect", "Training1"), _

          "--PLEASE_SELECT--"))

    Me.Year.Enabled = False
End Sub
Fill the Year Dropdown

The Year dropdown needs to be filled with unique Year names from the VehicleType table.  This function is similar to the Populate Make and Populate Model functions except with one difference.  The currently selected Make and Model are added to the Where Clause before retrieving new records from the database.

Note that in this function, the ListItem created will contain the Year name and the Vehicle Type Id as the value.  This will allow us to save the Vehicle Type Id in the Owner Vehicle table when a vehicle is selected.
Protected Overrides Sub PopulateYearDropDownList( _


ByVal selectedValue As String, _


ByVal maxItems As Integer)

    'Step 1: Setup the WHERE clause.
    Dim wc As WhereClause = CreateWhereClause_YearDropDownList()

    wc.iAND(VehicleTypeTable.Make, BaseFilter.ComparisonOperator.EqualsTo, Me.Make.SelectedValue)

    wc.iAND(VehicleTypeTable.Model, BaseFilter.ComparisonOperator.EqualsTo, Me.Model.SelectedValue)

    Dim orderBy As OrderBy = New OrderBy(False, True)

    orderBy.Add(VehicleTypeTable.Make, OrderByItem.OrderDir.Asc)

    Me.Year.Items.Clear()

    Dim rec As VehicleTypeRecord

    Dim pageNum As Integer = 0
    Dim recs As VehicleTypeRecord()

    ' Step 2: Read from the database and fill the Make dropdown

' until either there are no more records in the DB

' or until we have reached the maxItems
    Do
        recs = VehicleTypeTable.GetRecords(wc, orderBy, pageNum, maxItems)

        For Each rec In recs

            ' Create the item and add to the list.
            Dim item As ListItem = New ListItem(rec.Year, rec.Id0.ToString)

            ' If the item is already in the list, do not add it.
            If Not Me.Year.Items.Contains(item) Then
                Me.Year.Items.Add(item)

            End If
        Next
        pageNum += 1
    Loop While recs.Length = maxItems AndAlso Me.Year.Items.Count < maxItems

    ' Step 3: Setup the selected item.
    If Not selectedValue Is Nothing AndAlso _




selectedValue.Trim <> "" AndAlso _




Not SetSelectedValue(Me.Year, selectedValue) AndAlso _




Not MiscUtils.SetSelectedValue(Me.Year, OwnerVehicleTable.VehicleTypeId.Format(selectedValue)) Then
        Dim fvalue As String = OwnerVehicleTable.VehicleTypeId.Format(selectedValue)

        Dim item As ListItem = New ListItem(fvalue, selectedValue)

        item.Selected = True
        Me.Year.Items.Insert(0, item)

    End If
    Me.Year.Items.Insert(0, New ListItem(Page.GetResourceValue("Txt:PleaseSelect", "Training1"), _

           "--PLEASE_SELECT--"))

    ' Step 4: Disable & clear other dropdowns
End Sub
How it works

There are three dropdown shown for each row in the detail section of the Add Owner page.  The list of available makes are in the first dropdown.  

Once the make is selected by the user, the second dropdown shows the specific models belonging to the selected make.

And when the end user selects the make and the model, the available years are shown.

